```
> restart;
> with(DEtools):
```

Model of Maltus

Define the model of Maltus as a differential equation in Maple:

```
> maltus := diff( p(t),t ) = alpha * p(t);
```

$$maltus := \frac{d}{dt}\, p(t) = \alpha\, p(t)$$

Let Maple compute the general solution (note: the solution is given as an equation!)

```
> maltsol := dsolve({ maltus, p(0)= p0}, p(t));
```

$$maltsol := p(t) = p0\, e^{\alpha t}$$

Convert the solution into a function ('unapply' converts a term into a function):

```
> maltfun := unapply( subs( maltsol, p(t) ), t );
```

$$maltfun := t \rightarrow p0\, e^{\alpha t}$$

Specify initial value p0 and value for alpha, and plot solution -
here an example of a decaying population:

```
> maltex1 := subs( {p0 = 2, alpha = -0.5}, eval(maltfun) ):
  plot( maltex1, 0..10, p=0..3, thickness=3);
```



. . . and an example of a growing population

```
> maltex2 := subs( {p0 = 0.2, alpha = 0.3}, eval(maltfun) ):
  plot( maltex2, 0..10, p=0..3, thickness=3);
```



Model of Verhulst

Set up ode with a linear growth term:

```
> verhulst := diff(p(t),t) = alpha - beta *p(t);
```

$$verhulst := \frac{d}{dt}\, p(t) = \alpha - \beta\, p(t)$$

```
> verhsol := dsolve({ verhulst, p(0)= p0}, p(t));
```

$$verhsol := p(t) = \frac{\alpha}{\beta} + e^{-\beta t}\left(p0 - \frac{\alpha}{\beta}\right)$$

```
> verhfun := unapply( subs( verhsol, p(t)), t);
```

$$verhfun := t \rightarrow \frac{\alpha}{\beta} + e^{-\beta t}\left(p0 - \frac{\alpha}{\beta}\right)$$

First example: the saturation value is alpha/beta = 2, and the initial population is larger than the saturation value:

```
> verhex1 := subs( {alpha=1, beta=0.5, p0=3}, eval(verhfun) );
  plot(verhex1, 0..10, p=0..5, thickness=3);
```

$$verhex1 := t \rightarrow 2.000000000 + 1.000000000\, e^{-0.5\, t}$$



Second example: initial population is smaller than saturation value:

```
> verhex2 := subs( {alpha=1, beta=0.5, p0=1}, eval(verhfun) );
  plot(verhex2, 0..10, p=0..5, thickness=3);
```

$$verhex2 := t \rightarrow 2.000000000 - 1.000000000\, e^{-0.5\, t}$$



Logistic Growth

```
> logistic := diff(p(t),t) = alpha * ( 1 - p(t)/beta ) * p(t);
```

$$logistic := \frac{\mathrm{d}}{\mathrm{d}t}\, p(t) = \alpha \left(1 - \frac{p(t)}{\beta}\right) p(t)$$

```
> logisol := dsolve({ logistic, p(0)= p0}, p(t));
```

$$logisol := p(t) = \frac{p0\,\beta}{p0 + \mathrm{e}^{-\alpha t}\,\beta - \mathrm{e}^{-\alpha t}\,p0}$$

```
> logifun := unapply( subs( logisol, p(t)), t);
```

$$logifun := t \rightarrow \frac{p0\,\beta}{p0 + \mathrm{e}^{-\alpha t}\,\beta - \mathrm{e}^{-\alpha t}\,p0}$$

```
> logiex1 := subs( {alpha=1, beta=2, p0=0.1}, eval(logifun) );
  plot(logiex1, 0..8, p=0..3, thickness=3);
```

$$logiex1 := t \rightarrow \frac{0.2}{0.1 + 1.9\,\mathrm{e}^{-t}}$$




```
> logiex2 := subs( {alpha=1, beta=2, p0=3}, eval(logifun) );
  plot(logiex2, 0..8, p=0..3, thickness=3);
```

$$logiex2 := t \rightarrow \frac{6}{3 - e^{-t}}$$



Logistic Growth with Threshold

```
> threshold := diff(p(t),t) = alpha*( 1 - p(t)/beta ) * ( 1 -
  p(t)/delta) * p(t);
```

$$threshold := \frac{d}{dt}\, p(t) = \alpha \left(1 - \frac{p(t)}{\beta}\right) \left(1 - \frac{p(t)}{\delta}\right) p(t)$$

Maple can still solve it, but the solution becomes rather complicated:

```
> thrsol := dsolve({ threshold, p(0)= p0}, p(t));
```

$$thrsol := p(t) = e^{RootOf(\ln(e^{\_Z} + \beta)\, \beta - \ln(p0)\, \beta + \beta \ln(p0 - \delta) - \alpha\, t\, \beta - \beta \ln(-\delta + e^{\_Z} + \beta) + \_Z\, \delta + \alpha\, t\, \delta + \ln(p0)\, \delta - \ln(e^{\_Z} + \beta)\, \delta - \delta \ln(-\beta + p0))} + \beta$$

So we will tell Maple the values of alpha, beta, and delta earlier:

```
> threshold1 := subs( {alpha=-1, beta=2, delta=4}, threshold);
```

$$threshold1 := \frac{d}{dt}\, p(t) = -\left(1 - \frac{1}{2}\, p(t)\right) \left(1 - \frac{1}{4}\, p(t)\right) p(t)$$

```
> dsolve({ threshold1, p(0)=2.2}, p(t));
  threx1 := unapply( subs( %, p(t) ), t):
  plot(threx1, 0..8, p=0..5, thickness=3);
```

$$p(t) = -\frac{2\left(e^{t}+99-\sqrt{\left(e^{t}\right)^{2}+99\,e^{t}}\right)}{e^{t}+99}+4$$




```
> dsolve({ threshold1, p(0)=1.9}, p(t));
  threx2 := unapply( subs( %, p(t) ), t):
  plot(threx2, 0..8, p=0..5, thickness=3);
```

$$p(t) = -\frac{2\left(e^{t}+399+\sqrt{\left(e^{t}\right)^{2}+399\,e^{t}}\right)}{399+e^{t}}+4$$




```
> dsolve({ threshold1, p(0)=5}, p(t));
  threx3 := unapply( subs( %, p(t) ), t):
  plot(threx3, 0..8, p=0..5, thickness=3);
```

$$p(t) = 2\,e^{\frac{1}{2}t}\sqrt{\frac{1}{-1+e^{t}\,e^{2\ln\left(\frac{3}{5}\right)+\ln(5)}}}\;e^{\ln\left(\frac{3}{5}\right)+\frac{1}{2}\ln(5)}+2$$



5
4
3
2
1
0
p
0
1
2
3
4
5
6
7
8